
EE 308

Exam 1

February 15, 2002

Name:

You may use any of the Motorola data books, and the overheads posted on the Internet. No calculators allowed. Show
all work. Partial credit will be given. No credit will be given if an answer appears with no supporting work.

1. This problem deals with hex and binary arithmetic

(a) Fill in the blanks in this table. The hex and binary numbers are 8-bits.

Unsigned Signed
Hex Binary Decimal Decimal

A7 10100111 167 -89

AF 10101111 175 -81

51 01010001 81 81

(b) The following operations are done in accumulator A of an HC12. Indicate the answer in accumulator A,
and the state of the flags after the operations.

94 7F C7 75 4C
+ 6C + 1A + D3 - C9 - E6

Acc. A 00 99 9A AC 66
C 1 0 1 1 1
V 0 1 0 1 0
N 0 1 1 1 0
Z 1 0 0 0 0

2. Consider the following program fragment:

CODE section .text
org $0800
ldaa #$23 ; 1 cycle
adda $091A ; 3 cycles

loop: deca ; 1 cycle
cmpa $02 ; 3 cycles
bne loop ; 3/1 cycles
swi

(a) Hand-assemble the code fragment. That is, show what hex numbers will be in what memory locations
after this program has been assembled. (Note: There are more lines in the table below than you need.)

Address Byte

0800 86
0801 23
0802 BB
0803 09
0804 1A
0805 43
0806 91
0807 02
0808 26
0809 FB
080A 3F

(b) How many bytes of memory does the code fragment occupy?

11 Bytes (10 Bytes okay if you didn’t include the SWI)

(c) How many processor clock cycles will it take to execute the five instructions of the code fragment?

1 + 3 + 1 + 3 + 3/1 = 11/9 Either 11 or 9 is acceptable

(d) How many microseconds will it take the fragment to execute on an HC12 with an 8 MHz processor clock?

11 cycles x 0.125 us/cycle = 1.375 us
or
9 cycles x 0.125 us/cycle = 1.125 us

3. Below are some data in the HC12 memory:

0 1 2 3 4 5 6 7 8 9 A B C D E F

0900 D6 05 35 CF E0 00 FE 08 20 A6 00 47 6A 05 08 53
0910 26 F7 34 C6 C8 CD 9C 40 03 26 FD 53 26 F7 3D 3F
0920 07 C2 3A 68 F3 09 C2 67 9A 0F AA 55 08 40 CD CF

Using the above memory values, consider the instructions below.

� Indicate the starting address for each instruction. (Note that the first instruction is at address $0800.)

� Indicate the values in the registers (in hex) after the HC12 executes each of the instructions. (If the instruction does not change a register, you may
leave that entry blank.)

� Show the state of the N, Z, V and C condition code bits after each instruction has been executed.

� Indicate the addressing mode used by each instruction.

� Indicate the effective address of each instruction.

� Write down the number of cycles needed to execute each instruction.

D Addressing Effective Number
Address Instruction A B X Y SP N Z V C Mode Address of Cycles

AA 37 0910 0900 0920 1 0 1 1
$0800 ldy #$091A 091A 0 0 0 1 IMM 0801 2
$0803 ldy $091A FD53 1 0 0 1 EXT 091A 3
$0806 adda 5,X 77 0 0 1 1 IDX 0915 3
$0808 puly 07C2 0922 0 0 1 1 INH 0920 3
$0809 cmpb #$5A 1 0 0 1 IMM 080A 1
$080B ldaa 2,-X 08 090E 0 0 0 1 IDX 090E 3
$080D negb C9 1 0 0 1 INH none 1

4. An HC12 has been hooked up such that the 8-bit signed number stored at address $0072 represents the temper-
ature in this room in degrees Fahrenheit. Also, a switch which controls the heater for this room is connected to
Bit 0 of PORTA, and a switch which controls the air conditioner for this room is connected to Bit 4 of PORTA.
(Writing a 0 to Bit 0 turns the heater off; writing a 1 to Bit 0 turns the heater on. Writing a 0 to Bit 4 turns the
air conditioner off; writing a 1 to Bit 4 turns the air conditioner on.) Write a program for the HC12 which will
do the following:

� Set up Bits 0 and 4 of PORTA as output bits. The other bits of PORTA should be set up as inputs.
� If the temperature in the room is below 68 degrees, make sure the air conditioner is off and the heater is

on.
� If the temperature in the room is between 68 degrees and 75 degrees, make sure both the heater and air

conditioner are off.
� If the temperature in the room is above 75 degrees, make sure the air conditioner is on and the heater is

off.
� Repeat this set of instructions forever.

temp: equ $0072 ; Address of temperature register
PORTA: equ $0000 ; address of PORTA
DDRA: equ $0002 ; address of DDRA

low_temp: equ 68 ; below this, turn on heat
hi_temp: equ 75 ; above this, turn on air

heat_mask: equ 00000001b ; Bit 0 for heater
air_mask: equ 00010000b ; Bit 4 for air conditioner

prog: equ $0800

org prog
clr DDRA ; PORTA input

bset DDRA,#heat_mask ; make heat bit output
bset DDRA,#air_mask ; make air bit output

loop: ldaa temp ; read temp
cmpa #low_temp
blt heat_on ; below 68, turn heat on
cmpa #hi_temp
bgt air_on ; above 75, turn air on

both_off: bclr PORTA,#heat_mask ; between 68 and 75,
bclr PORTA,#air_mask ; turn heat and air off
bra loop

heat_on: bset PORTA,#heat_mask ; turn heat on
bclr PORTA,#air_mask ; turn air off
bra loop

air_on: bclr PORTA,#heat_mask ; turn heat off
bset PORTA,#air_mask ; turn air on
bra loop

